
Risolviamo il problema 5. Questo caso è stato preso come esempio da Steven H. 

Strogatz nel suo libro Nonlinear Dynamics and Chaos, Perseus Books Publishing, 

Reading (MA), 1994, pp. 155-159. L’autore immagina che la competizione avvenga 

tra conigli e pecore, per la stessa risorsa alimentare (l’erba) in disponibilità limitata, 

e trascura ogni ulteriore “complicazione” del modello: predatori, effetti stagionali, 

altre fonti di cibo… Rimangono due principali effetti da prendere in considerazione: 
 

i) ciascuna specie dovrebbe crescere fino alla propria capacità portante in assenza 

dell’altra specie; e infatti sia la prima equazione del sistema (4) quando y = 0, sia la 

seconda quando x = 0, si riducono alla forma dell’equazione (2). Notoriamente i 

conigli sono assai prolifici, sicché dovremmo assegnare loro un più alto tasso netto 

di crescita: diciamo 3 per i conigli e 2 per le pecore. 
 

ii) Quando i conigli e le pecore si incontrano, cominciano i problemi. Talvolta un 

coniglio riesce a mangiare, ma più spesso è spinto via da una pecora che vuol 

brucare l’erba proprio in quella zona. Possiamo ipotizzare che questi conflitti 

accadano con una frequenza proporzionale alla quantità di ciascuna popolazione: ad 

esempio, se le pecore raddoppiassero, la probabilità che un coniglio ne incontri una 

diverrebbe due volte più grande. Inoltre, assumiamo che tali conflitti riducano il 

tasso di crescita di entrambe le specie, ma che l’effetto sia più grave per i conigli. 

I coefficienti, d’altra parte arbitrari, sono quindi scelti per rispettare questo scenario. 
 

Ecco come cambia la situazione relativamente alle isocline nulle e ai quattro punti di 

equilibrio: 
 

 
 

Infatti, i valori dei parametri sono: a = 3, b = 1, c = 2, d = 2, e = 1, f = 1, e quindi, 

numericamente: a/b = 3, a/c = 3/2, d/e = d/f = 2; inoltre, af − cd = bd − ae = bf − ce 

= −1, per cui il quarto punto di equilibrio è (x
*
 = 1, y

*
 = 1). 

Ma che cosa succede altrove? Per scoprirlo usiamo Maple… 

restart: 

with(DEtools): 

a:=3: b:=1: c:=2: d:=2: e:=1: f:=1: 

LVcomp := diff(x(t), t) = x(t)*(a-b*x(t)-c*y(t)), 

          diff(y(t), t) = y(t)*(d-e*x(t)-f*y(t)): 

vars := x(t), y(t): 

DEplot({LVcomp}, {vars}, t = 0..8, x = 0..4, y = 0..3, title = 

  `Modello competitivo di Lotka-Volterra: x = conigli, y = pecore`, 

  colour = magnitude); 

 

 
 

Inquadriamo meglio il rettangolo che comprende i punti di equilibrio: 
 

DEplot({LVcomp}, {vars}, t = 0..8, x = 0..3, y = 0..2, title = 

  `Modello competitivo di Lotka-Volterra: x = conigli, y = pecore`, 

  colour = magnitude); 

 

 
 

Si intuisce che (1, 1) è un punto di equilibrio non stabile (è infatti un punto sella). 

Tracciamo alcune traiettorie, servendoci questa volta della possibilità di osservarle 

mentre si compiono, rispettando la velocità relativa. Gli stati iniziali delle ultime due 

traiettorie sono stati scelti ad hoc… 

DEplot({LVcomp}, {vars}, t = 0..8, x = 0..4, y = 0..4, 

 [[x(0)=0.1, y(0)=0.1], [x(0)=0, y(0)=4], [x(0)=1, y(0)=4], 

 [x(0)=4, y(0)=1], [x(0)=4, y(0)=4], [x(0)=4, y(0)=3], 

 [x(0)=0, y(0)=1], [x(0)=0.1, y(0)=0.6], 

 [x(0)=1.5, y(0)=1.33389], [x(0)=0.5, y(0)=0.614814]], 

 animate = true, numframes = 60, arrows = none, thickness = 2, 

 linecolour = [gold,red,magenta,cyan,black,green,blue,cyan,blue,red]); 

 

 
 

Alcune traiettorie si dirigono verso il punto (3, 0), altre verso (0, 2): “in mezzo” ci 

deve essere una particolare traiettoria che si dirige invece verso il punto sella (1, 1), 

ma ovviamente è difficile da evidenziare! Infatti, replicando l’ultimo comando con  

t=0..40 e numframes = 300, al termine dell’animazione si ottiene: 
 

 
 

Facciamo un’altra prova: 
 
DEplot({LVcomp}, {vars}, t = 0..30, x = 0..4, y = 0..3, 

 [[x(0)=0.001, y(0)=0.001], [x(0)=4, y(0)=2], [x(0)=3, y(0)=3], 

 [x(0)=0.001, y(0)=0], [x(0)=1, y(0)=2], [x(0)=0.2, y(0)=0.1], [x(0)=0.1, y(0)=0.6], 

 [x(0)=0.1, y(0)=0.2032], [x(0)=0.1, y(0)=0.2033], 

 [x(0)=2, y(0)=1.639], [x(0)=2, y(0)=1.640]], 

 animate = true, numframes = 225, arrows = none, thickness = 1, 

 linecolour = [gold,magenta,blue,black,green,cyan,cyan,red,black,red,blue]); 

Che cosa se ne deduce? 

Volendo osservare l’andamento nel tempo di una soluzione, diamo i comandi: 
 

DEplot({LVcomp}, {vars}, t = 0..15, [[x(0)=4, y(0)=2]], 

  numpoints = 600, scene = [t, x], thickness = 1, linecolour = blue); 
 

 
 

DEplot({LVcomp}, {vars}, t = 0..15, [[x(0)=4, y(0)=2]], 

  numpoints = 600, scene = [t, y], thickness = 1, linecolour = red); 

In conclusione: l’origine è un punto di equilibrio instabile (è una sorgente); (3, 0) e 

(0, 2) sono punti di equilibrio stabile; infine, come si è già detto, (1, 1) è un punto 

sella, e perciò non stabile, a meno di spostarsi proprio su una delle due traiettorie 

(dette separatrici) che costituiscono la cosiddetta varietà stabile (stable manifold, in 

inglese) della sella: una che parte “in prossimità” dell’origine e l’altra che arriva 

“dall’infinito”, ed entrambe vanno a finire nel punto sella (saddle point ). 

La seguente figura è tratta dal libro di Strogatz (op. cit., p. 159). 

 

 
 

Questo quadro di stato ha un’interessante interpretazione biologica: mostra infatti 

che, in generale, una delle due specie porta l’altra, prima o poi, all’estinzione. Le 

traiettorie che partono al di sotto della varietà stabile della sella conducono alla 

futura estinzione delle pecore: nel loro complesso, esse costituiscono il bacino di 

attrazione del punto di equilibrio (3, 0), evidenziato in grigio nella figura. In tale 

stato esisteranno soltanto conigli. Viceversa, le traiettorie che partono al di sopra 

della varietà stabile della sella portano alla futura estinzione dei conigli, e nel loro 

complesso formano il bacino di attrazione del punto di equilibrio (0, 2), stato in cui 

esisteranno soltanto pecore. (In generale, i bacini di attrazione e i loro “confini”, 

cioè le traiettorie separatrici, rivestono notevole importanza perché suddividono lo 

spazio delle fasi in regioni caratterizzate da diversi comportamenti a lungo termine.)  

Questa dicotomia si ritrova in altri modelli di competizione e ha indotto i biologi a 

formulare il principio di esclusione competitiva: due specie, che occupano la stessa 

“nicchia ecologica” e competono per la stessa risorsa limitata, non possono 

coesistere, vale a dire che una delle due è destinata ad estinguersi. 

Nel caso del modello (4), occupare la stessa nicchia significa sostanzialmente che 

c / b = f / e (da cui bf − ce = 0), poiché questa uguaglianza implica che l’effetto 

competitivo di y su x sia paragonabile a quello di y su sé stesso; ne vedremo un paio 

di esempi risolvendo il prossimo problema. Invece, nell’esempio sopra studiato, 

questa uguaglianza non sussiste; tuttavia, è estremamente difficile che, qualora lo 

stato venga a trovarsi nel punto sella (non stabile), vi possa rimanere a lungo! 


